

Buffer Granulation



different approaches to buffer granulation with gui examples

Description

As with many things in SC work can be done by language or server. Speaking about granulation in general this mainly concerns the control of single grains, their rate, timing, length and other parameters. Regarding buffer granulation specifically this task can e.g. be taken over by ugens like TGrains and GrainBuf, the latter additionally accepting a grain envelope arg. The SC plugin distribution contains further variations (see BhobUGens, JoshUGens). By using granulation ugens in a SynthDef granular textures can be produced with a single Synth, including grain parameter sequencing with demand rate ugens, the [DX suite](#) opens additional genuine options in this regard. Alternatively SynthDefs for single grains can be defined with PlayBuf or BufRd in order to control the whole buffer granulation process from language side, using Patterns, Tasks or Routines. What is the best way? To a large extent this is a question of personal preference. Regarding CPU performance granulation ugens have an advantage, whereas e.g. pattern-driven granulation allows concise control over the sequencing of granulation parameters in a clear syntax. If pattern-driven granulation is becoming CPU-critical you might want to consider the event type `\grain`, a lightweight variant of default type `\note` (see comment in the source file `Event.sc`). For granulation with Tasks see [VarGui, Ex.3](#).

Also hybrid strategies are possible, e.g. language controlled setting of a single granulation Synth or involving extra control synths in a language-driven granulation process, further options are Pspawner / Pspawn and Wavesets. VarGui can be used to integrate these setups in single GUIs. Together with this file (`miSCellaneous v0.7`) I reinvented a color grouping option that overrides automatic color grouping ([VarGui, Ex.7](#)). The latter is based on the logical structure of ordinary and array controls, synths and environments. This is useful for VarGuIs up to a medium number of controls per Synth / Pattern / Task and also for a large number of such items, but it doesn't handle cases well where there are many controls per item. Not barely an aesthetic detail, it is much more convenient for experimenting to group all sliders of a Synth that, say, have to do with a bandpass filter, within one color.

Types of variables

In general interpreter variables are preferred in examples below, wherever possible. If evaluated with example code only their values are passed and further changing of used variables doesn't affect already generated gui instances. On the contrary repeated evaluation of an interpreter variable, e.g. from a Pfunc, is unsafe - variable's value could change while gui has not yet been closed - so in concerned examples values are passed to the event definition, in [Ex.2c](#) a variable is declared for the same reason. The use of environmental variables is following the way VarGui is handling them. Per default every EventStreamPlayer derived from a passed Pattern is run in a separate newly generated Environment, where variables are being set and Streams from Pfuncs and PLx patterns are reading from. See [Event patterns and Functions](#), [PLx suite](#) and [VarGui](#).

WARNING:

1. Be careful with amplitudes, especially with buffers you haven't granulated before! Also keep in mind that a granular cloud moving through a buffer can suddenly become louder and other controls than amp (e.g. buffer position, trigger rate, bandpass parameters) can cause a raise of amplitude too.
2. I haven't used below setups for live performances. Although all of them work stable for me as they are, in general hangs can occasionally happen with pattern-driven setups. Often this can be tracked down to sequences of extremely short event durations (and/or long grain durations). Where this can happen as a side-effect, thresholds can be built in, e.g. [Ex.2d](#) (Wavesets) has a parameter `maxTrigRate`. Another possible source of hangs is careless deep nesting of Patterns where mistakes can easily occur. Starting with clear Pattern structures is recommended - and if more complications are involved: testing without sound first, after saving your patch, might be a good idea.

NOTE: All variants from this tutorial can be applied to a buffer, which is occasionally (or continuously) filled with live input. Vice versa variants from [Live Granulation](#) can of course be applied to any signal, thus also to any playback of a buffer.

All examples below expect mono buffers. Buffer paths referring to the included sample suppose that you have installed via quarks or moved `miSCellaneous lib` into the user or system extensions directory directly (not into a subfolder), if not so or you have moved the sample file somewhere else you'd have to change paths accordingly.

Due to a bug in SC 3.7 / 3.8 TGrains didn't response to amp changes, I changed the examples accordingly, the bug is fixed in 3.9.

While other parts of `miSCellaneous lib` were running fine I was unable to allocate buffers with the SC 3.5.4 Windows binary in August 2012. Meanwhile this issue has been solved (SC 3.6.6, February 2014). There also seemed to be accuracy issues with `OffsetOut` on Windows with SC 3.6 still, but not with newer versions.

Credits

Thanks for contributions and inspirations by SCers Alberto de Campo (Wavesets), James Harkins (Patterns), Ron Kuivila (Pspawner), Sergio Luque (stochastic distributions), Josh Parmenter (granulation plugins), Bhob Rainey (granulation plugins).

References

1. de Campo, Alberto. "Microsound" In: Wilson, S., Cottle, D. and Collins, N. (eds). 2011. The SuperCollider Book. Cambridge, MA: MIT Press, 463-504.
2. Luque, Sergio (2006). Stochastic Synthesis, Origins and Extensions. Institute of Sonology, Royal Conservatory, The Netherlands. <http://sergioluque.com>
3. Roads, Curtis (2001). Microsound. Cambridge, MA: MIT Press.
4. Seidl, Fabian (2016). Granularsynthese mit Wavesets für Live-Anwendungen. Master Thesis, TU Berlin. http://www2.ak.tu-berlin.de/~akgroup/ak_pub/abschlussarbeiten/2016/Seidl_MasA.pdf

5. Wishart, Trevor (1994). Audible Design. York: Orpheus The Pantomime Ltd.
6. Xenakis, Iannis (1992). Formalized Music. Hillsdale, NY: Pendragon Press, 2nd Revised edition.

1. Granulation with Ugens

Ex.1a: Basic buffer granulation Synth

```

1 (
2 s = Server.local;
3 Server.default = s;
4 s.boot;
5 )
6
7 // basic SynthDef suited for pitch-shift and time-stretch
8 // buffer position given relatively (posLo and posHi between 0 and 1)
9 // posDev: maximum amount of deviation from (moving) grain center position
10 // posDev = 0 can lead to comb filter effects (which may be nice sometimes)
11
12 // passing control specs as metadata allows for VarGui shortcut build method sVarGui
13 // metadata specs can be overwritten by arg ctrReplace
14 // alternatively control specs may be passed as synthCtr arg to a build with VarGui(
15 ... )
16 (
17 SynthDef(\gran_1a, { arg out = 0, bufNum = 0, posLo = 0.0, posHi = 1.0,
18   posRate = 1, posDev = 0.01, trigRate = 100, granDur = 0.1, rate = 1.0,
19   panMax = 1, amp = 0.1, interp = 4;
20
21   var trig, pan, pos, bufDur, bufDurSection, posDif;
22
23   posDif = posHi - posLo;
24   bufDur = BufDur.kr(bufNum);
25   bufDurSection = bufDur * posDif;
26   trig = Impulse.kr(trigRate);
27   pos = posLo * bufDur +
28     (Phasor.ar(0, BufRateScale.kr(bufNum) * posRate / SampleRate.ir, posLo *
29   bufDur, posHi * bufDur) +
30   (TRand.kr(-0.5 * posDev, 0.5 * posDev, trig) * bufDur)).mod(bufDurSection);
31   pan = Demand.kr(trig, 0, Dseq([panMax, panMax.neg], inf) * 0.999);
32   Out.ar(out, TGrains.ar(2, trig, bufNum, rate, pos, granDur, pan, 1, interp) *
33   amp);
34   }, metadata: (
35     specs: (
36       posLo: [0.01, 0.99, \lin, 0.01, 0],
37       posHi: [0.01, 0.99, \lin, 0.01, 1],
38       posRate: [0.1, 2, \lin, 0.01, 1],
39       posDev: [0, 0.2, 5, 0, 0.01],
40       granDur: [0.01, 0.3, \lin, 0.01, 0.1],
41       trigRate: [1, 200, \lin, 0.01, 100],
42       rate: [0.1, 2, \lin, 0.01, 1],
43       panMax: [0.0, 1, \lin, 0.005, 0.8],
44       amp: [0.0, 0.5, \lin, 0.005, 0.25]
45     )
46   )
47 )
48 ).add;
49

```

```

47 b = Buffer.read(s, Platform.miSCellaneousDirs[0] ++ "Sounds" ++
    "kitchen_sounds_1.wav");
48 // This searches the most likely extension places for the miSCellaneous folder.
49 // In case of an extraordinary install situation or a removed sound file, pass the
    concerned path.
50
51 )
52
53
54 // start from GUI
55
56 \gran_1a.sVarGui([\bufNum, b.bufnum]).gui;

```

Ex.1b: More deviations

```

1 // In example 1a only a deviation from the grain center position was implemented.
2 // With additional deviation controls a greater plasticity of sound can be achieved,
3 // here deviations are added for trigRate (LFO with oscillation freq and deviation
    max),
4 // grain duration and rate (TRand, equally weighted random deviation with given max).
5 // Deviations intervals could be defined alternatively,
6 // e.g. (1/(1+maxDev), 1+maxDev) with 0 < maxDev
7 // instead of (1-maxDev, 1+mexDev) with 0 < maxDev < 1
8
9 // posRate control range is widened by inventing two controls for
10 // mantissa and exponent, so posRate = 1 for init param pair
11 // posRateE = 0 and posRateM = 1
12
13 (
14 SynthDef(\gran_1b, { arg out = 0, bufNum = 0, posLo = 0.0, posHi = 1.0,
15   posRateE = 0, posRateM = 1, posDev = 0.01, trigRate = 100, trigRateDev = 0,
16   trigRateOsc = 1, granDur = 0.1, granDurDev = 0, rate = 1.0, rateDev = 0,
17   panMax = 1, amp = 0.1, interp = 4;
18
19   var trig, pan, pos, bufDur, bufDurSection, posDif, posRate;
20
21   posDif = posHi - posLo;
22   bufDur = BufDur.kr(bufNum);
23   bufDurSection = bufDur * posDif;
24   trig = Impulse.kr(LFDNoise3.kr(trigRateOsc, trigRate * trigRateDev, trigRate));
25   posRate = 10 ** posRateE * posRateM;
26   pos = posLo * bufDur +
27     (Phasor.ar(0, BufRateScale.kr(bufNum) * posRate / SampleRate.ir, posLo *
    bufDur, posHi * bufDur) +
28     (TRand.kr(-0.5, 0.5, trig) * posDev * bufDur)).mod(bufDurSection);
29   pan = Demand.kr(trig, 0, Dseq([panMax, panMax.neg], inf) * 0.999);
30   Out.ar(out, TGrains.ar(2, trig, bufNum, rate * (TRand.kr(-1, 1.0, trig) * rateDev
    + 1), pos,
31     granDur * (TRand.kr(-1, 1.0, trig) * granDurDev + 1), pan, 1, interp) * amp);
32   }, metadata: (
33     specs: (
34       posLo: [0.01, 0.99, \lin, 0.01, 0],
35       posHi: [0.01, 0.99, \lin, 0.01, 1],
36       posRateE: [-3, 4, \lin, 1, 0],
37       posRateM: [0.1, 10, \exp, 0.01, 1],
38       posDev: [0, 0.2, 5, 0, 0.05],
39       trigRate: [1, 200, \lin, 0.01, 100],
40       trigRateDev: [0.0, 1, \lin, 0.01, 0],

```

```

41     trigRateOsc: [0.1, 2, \lin, 0.01, 3],
42     granDur: [0.01, 0.3, \lin, 0.01, 0.1],
43     granDurDev: [0.0, 0.95, \lin, 0.01, 0],
44
45     rate: [0.1, 2, \lin, 0.01, 1],
46     rateDev: [0.0, 0.99, \linear, 0.01, 0.05],
47     panMax: [0.0, 1, \lin, 0.005, 0.8],
48     amp: [0.0, 0.5, \lin, 0.005, 0.25]
49   )
50 )
51 ).add;
52
53 b = Buffer.read(s, Platform.miSCellaneousDirs[0] ++ "Sounds" ++
54 "kitchen_sounds_1.wav");
55 // This searches the most likely extension places for the miSCellaneous folder.
56 // In case of an extraordinary install situation or a removed sound file, pass the
57 // concerned path.
58 )
59 // start from GUI
60 // use color grouping for better overview
61
62 \gran_1b.sVarGui([\bufNum, b.bufnum]).gui(synthColorGroups:
63 (0..14).clumps([1,5,3,2,2,1,1]) )

```

Ex.1c: Buffer granulation Synth with external control synths

```

1 // This is a more modular approach, once having a basic granulation synth
2 // it can be linked with arbitrary control synths, here
3 // also moving through the buffer is controlled externally.
4 // Depending on LFOs it might be worth defining
5 // audio buses for control to get higher precision.
6
7
8 (
9 SynthDef(\gran_1c, { arg out = 0, bufNum = 0, amp = 0.1, pos = 0.5, posDev = 0.01,
10   trigRate = 100, granDur = 0.1, rate = 1, panMax = 1, interp = 4;
11   var trig, pan;
12   trig = Impulse.kr(trigRate);
13   pan = Demand.kr(trig, 0, Dseq([panMax, panMax.neg],inf) * 0.999);
14   pos = (pos * BufDur.kr(bufNum) * WhiteNoise.kr(posDev, 1));
15   Out.ar(out, TGrains.ar(2, trig, bufNum, rate, pos, granDur, pan, 1, interp) *
16   amp);
17   }).add;
18
19 // LFO synthdef for switching between 3 types:
20 // 0: LFDNoise3 (smooth random movement)
21 // 1: SinOsc
22 // 2: LFSaw (works as phasor for position)
23
24 SynthDef(\lfo, { |out = 0, lfoType = 0, freq = 1, lo = 0, hi = 1|
25   var ctrl;
26   ctrl = Select.kr(lfoType, [
27     LFDNoise3,

```

```

28     SinOsc,
29     LFSaw
30     ].collect { |x| x.kr(freq, mul: hi-lo/2, add: hi+lo/2) });
31     Out.kr(out, ctrl);
32   }).add;
33
34   // multichannel control bus
35   c = Bus.control(s, 5);
36
37   b = Buffer.read(s, Platform.miSCellaneousDirs[0] ++ "Sounds" ++
38     "kitchen_sounds_1.wav");
39   // This searches the most likely extension places for the miSCellaneous folder.
40   // In case of an extraordinary install situation or a removed sound file, pass the
41   // concerned path.
42   )
43
44   // In contrast to Ex. 1a and 1b VarGui is called explicitly
45   // as control specs should differ.
46
47   // Also here the granulation Synth is generated explicitly and not by the interface
48   // as its controls have to be mapped to buses
49   (
50     // start granulation synth paused and register
51     // to let VarGui know its state
52
53     x = Synth.newPaused(\gran_1c, [\bufNum, b]).register;
54
55     // map args to be controlled to consecutive subbuses
56     x.map(*([\pos, \trigRate, \granDur, \rate, \panMax],
57       (0..4).collect(c.subBus(_))).flop.flat)
58   )
59   (
60     // Open VarGui interface, granulation synth (#5) is paused
61     // (orange button on yellow background).
62     // Shift-clicking one of the green buttons runs
63     // granulation synth and control synths (#0 - #4).
64     // All synths can be paused (orange button) and resumed.
65
66     // NOTE: When stopping the granulation synth
67     // the linkage with control synths is lost!
68     // A new synth generated by the interface
69     // (by pressing the blue button of #5) is not
70     // automatically mapped to control buses.
71     // On the other hand control synths might be stopped and
72     // newly generated.
73
74
75   VarGui(synthCtr: [[
76     \pos, 0, // dummy spec for labelling
77     \out, c.index,
78     \lfoType, [0, 2, \lin, 1, 0],
79     // as LFOs are unified, original movement tempo is
80     // indicated not by 1 but by 1 / buffer duration
81     \freq, [0.01, 20, \exp, 0.0, 1/b.duration],
82     \lo, [0.0, 1, \lin, 0, 0.1],
83     \hi, [0.0, 1, \lin, 0, 0.9]
84   ], [
85     \trigRate, 1,
86     \out, c.index + 1,
87     \lfoType, [0, 2, \lin, 1, 0],
88     \freq, [0.001, 0.5, \exp, 0.0, 0.2],

```

```

89     \lo, [1, 100, \lin, 0, 7],
90     \hi, [1, 100, \lin, 0, 40]
91 ],[
92     \granDur, 2,
93     \out, c.index + 2,
94     \lfoType, [0, 2, \lin, 1, 1],
95     \freq, [0.001, 0.5, \exp, 0.0, 0.4],
96     \lo, [0.01, 0.2, \lin, 0, 0.1],
97     \hi, [0.01, 0.2, \lin, 0, 0.15]
98 ],[
99     \rate, 3,
100    \out, c.index + 3,
101    \lfoType, [0, 2, \lin, 1, 0],
102    \freq, [0.001, 0.1, \exp, 0.0, 0.03],
103    \lo, [0.1, 3, \lin, 0, 0.6],
104    \hi, [0.1, 3, \lin, 0, 1.1]
105 ],[
106    \panMax, 4,
107    \out, c.index + 4,
108    \lfoType, [0, 2, \lin, 1, 0],
109    \freq, [0.001, 0.5, \exp, 0.0, 0.5],
110    \lo, [0.0, 1, \lin, 0, 0.06],
111    \hi, [0.0, 1, \lin, 0, 0.9]
112 ],[
113    \out, 0,
114    \bufNum, b.bufnum,
115    \posDev, [0.001, 0.2, \exp, 0, 0.02],
116    \amp, [0.0, 2, \lin, 0, 0.5]
117 ]],
118 // 5 lfo synths are generated by the interface (synthdef name passed)
119 // but granulation Synth is passed directly as object
120 synth: \lfo!5 ++ x
121 ).gui(sliderPriority: \synth, playerPriority: \synth);
122 )

```

Ex.1d: Buffer granulation Synth with demand rate ugens

```

1 // Repeated grain triggering within a synth can be defined by demand rate ugens,
2 // which is comfortable in connection with granular ugens.
3
4 // E.g. with TGrains you can trigger parameters like grain duration, playback rate,
5 // position, panning and amp, single grains will keep their params if they overlap.
6 // (Note that in the below implementation changes of demand rate array fields will
7 // apply
8 // also not before the next call of those fields in the synth)
9
10 // More refined per-grain control, e.g. per-grain filtering with filter parameter
11 // streams,
12 // can be done by using a multichannel trick, see Ex. 1e.
13
14 (
15 // length of demand rate sequence, you might want to check larger sizes
16 // in connection with gui arg tryColumnNum > 1
17 ~n = 5;
18
19 // posRate control range is widened by inventing two controls for
20 // mantissa and exponent, so posRate = 1 for init param pair
21 // posRateE = 0 and posRateM = 1
22
23 SynthDef(\gran_1d, { |out = 0, soundBuf, posLo = 0.1, posHi = 0.3,

```



```

22     posRateE = 0, posRateM = 1, granDurMul = 1, rateMul = 1, panMul = 1, amp = 0.5,
    interp = 2 |
23
24     var signal, bufDur, granDur, granGate, relGranDurs, pos, overlap, overlaps,
    overlapSeq,
25         pan, rate, relRates, rateSeq, posRate, granDurSeq;
26
27     // array args for demand rate sequencing, short form of NamedControl
28     relGranDurs = \relGranDurs.kr(0.1!~n);
29     relRates = \relRates.kr(1!~n);
30     overlaps = \overlaps.kr(1!~n);
31
32     // Dstutter (or Dunique) necessary as granDurSeq is polled twice: granGate and
    granDur
33     granDurSeq = Dstutter(2, Dseq(relGranDurs, inf));
34     rateSeq = Dseq(relRates, inf);
35     overlapSeq = Dseq(overlaps, inf);
36
37     granGate = TDuty.ar(granDurSeq * granDurMul);
38     granDur = Demand.ar(granGate, 0, granDurSeq * granDurMul);
39     rate = Demand.ar(granGate, 0, rateSeq) * rateMul;
40     pan = Demand.ar(granGate, 0, Dseq([1, -1], inf)) * 0.999 * panMul;
41     overlap = Demand.ar(granGate, 0, overlapSeq);
42
43     bufDur = BufDur.kr(soundBuf);
44     posRate = 10 ** posRateE * posRateM;
45
46     pos = Phasor.ar(0, BufRateScale.kr(soundBuf) * posRate / (SampleRate.ir *
    bufDur), posLo, posHi);
47     signal = TGrains.ar(2, granGate, soundBuf, rate, pos * bufDur, granDur * overlap,
    pan, 1, interp);
48
49     Out.ar(out, signal * amp);
50 }
51 ).add;
52
53 b = Buffer.read(s, Platform.miSCellaneousDirs[0] ++ "Sounds" ++
    "kitchen_sounds_1.wav");
54 // This searches the most likely extension places for the miSCellaneous folder.
55 // In case of an extraordinary install situation or a removed sound file, pass the
    concerned path.
56 )
57
58 (
59 // relGranDurs are multiplied with granDurMul, analogously relRates with rateMul.
60 // Changes of these params apply immediately in contrast to the array args
61 // used by demand rate ugens which are used with next demand.
62
63 // check out moving a number of sliders of one array by using
64 // Shift, Alt + Shift and Ctrl + Shift, see VarGui help Ex. 1c.
65
66
67 VarGui(synthCtr: [
68     soundBuf: b.bufnum,
69
70     posLo: [0, 1, \lin, 0, 0.1],
71     posHi: [0, 1, \lin, 0, 0.9],
72     posRateE: [-3, 4, \lin, -1, 0],
73     posRateM: [0.1, 10, \exp, 0.01, 0.5],
74
75     // generating control specifications depending on index
76     relGranDurs: { |i| [0.01, 0.1, \lin, 0, i * 0.005 + 0.02] } ! ~n,
77     granDurMul: [0.03, 2, \lin, 0, 0.25],
78     overlaps: [0.1, 5, \lin, 0, 1.5] ! ~n,

```



```

79
80   relRates: { |i| [0.1, 1.5, \lin, 0, (5-i) * 0.1 + 0.5] } ! ~n,
81   rateMul: [0.1, 2, \lin, 0, 0.5],
82
83   panMul: [0, 1, \lin, 0, 0.6],
84   amp: [0.0, 3, \lin, 0, 2]
85 ],
86   synth: \gran_1d
87 ).gui(
88   tryColumnNum: 1,
89   // color grouping for better overview
90   synthColorGroups: (0..(~n*3+8)).clumps([1,4,~n+1,~n,~n+1,1,1]),
91   labelWidth: 90,
92   sliderWidth: 280
93 );
94 )

```

Ex.1e: Buffer granulation Synth with per-grain effect processing (TGrains)

```

1 // TGrains and other granular ugens allow per-grain processing
2 // for a limited number of parameters (pos, rate etc.)
3 // Nevertheless it is possible to apply arbitrary effects with
4 // per-grain parameter changes, even if grains overlap.
5 // This can be achieved by defining the granulation output
6 // as a multichannel signal and appropriate triggering of fx parameters.
7 // The example is adapted from a recommendation of Julian Rohrerhuber.
8
9 // The method is elegant but also a bit tricky in terms of
10 // multichannel triggering and channel routing.
11 // See Ex.1f for achieving the same with DX ugens
12
13 (
14 // the multichannel size and equivalently:
15 // the maximum number of overlapping grains that might get
16 // different fx parameters have to be fixed.
17 // For convenience of later L/R-spatialization we take an
18 // even number ~n = 2 * ~m
19
20 ~m = 5;
21 ~n = 2 * ~m;
22
23 SynthDef(\gran_1e, { |out = 0, soundBuf, posLo = 0.1, posHi = 0.9,
24   posRateE = 0, posRateM = 1, rate = 1, panMax = 0.8, bpRQ = 0.1, bpLo = 50, bpHi =
25   5000,
26   amp = 1, bpFund = 100, overlap = 2, trigRate = 1, interp = 2|
27   var sig, sigL, sigR, bpFreq, chan, bpFreqSeqs, dUgen,
28     trig, trigs, bufDur, pos, posRate;
29
30   trig = Impulse.ar(trigRate);
31   // we need a multichannel trigger that steps through all consecutive channels
32   trigs = { |i| PulseDivider.ar(trig, ~n, ~n-1-i) } ! ~n;
33
34   chan = Demand.ar(trig, 0, Dseq((0..~n-1), inf));
35
36   posRate = 10 ** posRateE * posRateM;
37   bufDur = BufDur.kr(soundBuf);

```

```

37   pos = Phasor.ar(0, BufRateScale.kr(soundBuf) * posRate * SampleDur.ir / bufDur,
38   posLo, posHi);
39   sig = TGrains.ar(~n, trig, soundBuf, rate, pos * bufDur, overlap/trigRate,
40   // Panning convention is that from PanAz,
41   // speakers should be from 0 to 2, but (orientation)
42   // 1/n has to be subtracted for n speakers.
43   // If this isn't done correctly grains are spread onto more than one channel
44   // and per-grain application of fxs fails.
45   chan.linlin(0, ~n-1, -1/~n, (2*~n - 3)/~n), 1, interp);
46
47   dUgen = Dwhite(0.0, 1);
48   sig = sig.collect { |ch, i|
49   // this is the place to define fxs per channel/grain
50   // multichannel trigger is polling from a single demand ugen
51   bpFreq = Demand.ar(trigs[i], 0, dUgen).linlin(0, 1, bpLo, bpHi);
52
53   // amplitude compensation for lower rq of bandpass filter
54   BPF.ar(ch, bpFreq, bpRQ, (bpRQ ** -1) * (400 / bpFreq ** 0.5));
55   };
56
57   // routing to two channels ...
58   sigL = Mix(((0..(~m-1)) * 2).collect(sig[_]));
59   sigR = Mix(((0..(~m-1)) * 2 + 1).collect(sig[_]));
60
61   // ... in order to have L/R-spreading with panMax as in other examples
62   Out.ar(0, Pan2.ar(sigL, panMax.neg) + Pan2.ar(sigR, panMax) * amp)
63 }.add;
64
65
66 b = Buffer.read(s, Platform.miSCellaneousDirs[0] ++ "Sounds" ++
67 "kitchen_sounds_1.wav");
68 // This searches the most likely extension places for the miSCellaneous folder.
69 // In case of an extraordinary install situation or a removed sound file, pass the
70 // concerned path.
71 )
72 (
73 VarGui(synthCtr: [
74   soundBuf: b.bufnum,
75   posLo: [0, 1, \lin, 0, 0.2],
76   posHi: [0, 1, \lin, 0, 0.5],
77   posRateE: [-3, 4, \lin, -1, -1],
78   posRateM: [0.1, 10, \exp, 0.01, 0.8],
79   overlap: [0.1, ~n, \lin, 0, 12],
80   trigRate: [1, 100, \lin, 0, 45],
81   rate: [0.1, 2, \lin, 0, 1],
82
83   bpRQ: [0.05, 1, \lin, 0, 0.25],
84   bpLo: [50.0, 5000, \exp, 0, 50],
85   bpHi: [50.0, 5000, \exp, 0, 5000],
86   panMax: [0, 1, \lin, 0, 0.85],
87   amp: [0.0, 3, \lin, 0, 1]
88 ],
89   synth: \gran_1e
90 ).gui(
91   tryColumnNum: 1,
92   synthColorGroups: (0..12).clumps([1,4,2,1,3,1,1])
93 );
94 )

```

Ex.1f: Buffer granulation Synth with per-grain effect processing (DXEnvFan)

```

1 // DXEnvFan generates a multichannel envelope which can be used as trigger for
  granulation and fxs on grains.
2 // It encapsulates the trigger logic, which has been used explicitly in Ex. 1e.
3 // DX ugens can be used for a variety of microsound techniques, see their help files.
4
5 (
6 ~maxOverlap = 12;
7 a = Bus.audio(s, ~maxOverlap);
8
9 // overlap only settable in SC versions >= 3.9
10
11 SynthDef(\gran_1f, { |out = 0, soundBuf, bus = 0, posLo = 0.1, posHi = 0.9,
12   posRateE = 0, posRateM = 1, overlap = 2, trigRate = 1, rate = 1,
13   bpRQ = 0.1, bpLo = 50, bpHi = 5000, panMax = 0.8, amp = 1|
14   var sig, bpFreq, dUgen, bufDur, pos, posRate, playbuf, env, maxOverlap =
~maxOverlap;
15
16   posRate = 10 ** posRateE * posRateM;
17   bufDur = BufDur.kr(soundBuf);
18   pos = Phasor.ar(0, BufRateScale.kr(soundBuf) * posRate * SampleDur.ir / bufDur,
posLo, posHi);
19
20   // multichannel trigger
21   env = DXEnvFan.ar(
22     Dseq((0..maxOverlap-1), inf),
23     trigRate.reciprocal,
24     size: maxOverlap,
25     maxWidth: maxOverlap,
26     width: (Main.versionAtLeast(3, 9)).if { overlap }{ 2 },
27     // option to avoid unwanted triggers
28     zeroThr: 0.002,
29     // take equalPower = 0 for non-squared sine envelopes
30     // more efficient with helper bus
31     equalPower: 0,
32     bus: a
33   );
34   // multichannel playback, pos is triggered for each grain
35   playbuf = PlayBuf.ar(1, soundBuf, rate, env, pos * BufFrames.ir(soundBuf), 1);
36
37   dUgen = Dwhite(0, 1);
38   // multichannel trigger used to poll values from drate ugen
39   bpFreq = Demand.ar(env, 0, dUgen).linlin(0, 1, bpLo, bpHi);
40
41   // generate grains by multiplying with envelope
42   sig = playbuf * env;
43
44   // different frequency on each grain channel
45   sig = BPF.ar(sig, bpFreq, bpRQ, (bpRQ ** -1) * (400 / bpFreq ** 0.5));
46
47   // generate array of 5 stereo signals
48   sig = Pan2.ar(sig, Demand.ar(env, 0, Dseq([-1, 1], inf) * panMax));
49
50   // mix to out
51   Out.ar(0, Mix(sig) * amp)
52 }, metadata: (
53   specs: (
54     posLo: [0.01, 0.99, \lin, 0.01, 0],
55     posHi: [0.01, 0.99, \lin, 0.01, 0.5],

```

```

56     posRateE: [-3, 4, \lin, 1, -1],
57     posRateM: [0.1, 10, \exp, 0.01, 1.35],
58     trigRate: [1, 200, \lin, 0.01, 90],
59     overlap: [0.2, 12, \lin, 0.01, 7],
60     rate: [0.1, 2, \lin, 0.01, 0.75],
61     panMax: [0.0, 1, \lin, 0.005, 0.75],
62     bpLo: [100, 5000, \lin, 0, 300],
63     bpHi: [100, 5000, \lin, 0, 3000],
64     bpRQ: [0.05, 1, \lin, 0, 0.18],
65     amp: [0.0, 3, \lin, 0.005, 1]
66 )
67 ).add;
68
69 b = Buffer.read(s, Platform.miSCellaneousDirs[0] ++ "Sounds" ++
70 "kitchen_sounds_1.wav");
71 // This searches the most likely extension places for the miSCellaneous folder.
72 // In case of an extraordinary install situation or a removed sound file, pass the
73 // concerned path.
74 )
75 (
76 \gran_1f.sVarGui([\soundBuf, b.bufnum]).gui(
77     tryColumnNum: 1,
78     synthColorGroups: (0..12).clumps([1,4,2,1,3,1,1])
79 )

```

Ex.1g: Buffer granulation with (half) wavesets: ZeroXBufRd

```

1 // movement through the buffer of zero crossings
2 // with a slow pos rate we get repetitions of half wavesets
3
4 // load sound buffer
5
6 // This searches the most likely extension places for the miSCellaneous folder.
7 // In case of an extraordinary install situation or a removed sound file, pass the
8 // concerned path.
9
10 b = Buffer.read(s, Platform.miSCellaneousDirs[0] ++ "Sounds" ++
11 "kitchen_sounds_1.wav");
12
13 // allocate zeroX buffer
14 (
15 z = Buffer.alloc(s, b.duration * 44100 / 5, 1);
16 s.scope;
17 )
18
19
20 // write zero crossings, but no need to overwrite sound buffer
21 // this is caused by setting adjustZeroXs to -1
22
23 (
24 {
25     // use LeakDC to avoid extremely long half wavesets
26     var src = LeakDC.ar(PlayBuf.ar(1, b, BufRateScale.ir(b)));

```

```

27   ZeroXBufWr.ar(src, b, z, adjustZeroXs: -1, doneAction: 2);
28 }.play
29 )
30
31
32 // instead adjust from lang and get zeroXs
33
34 b.adjustZeroXs(z, { |z| ~zeroXs = z.reject(_==0) })
35
36 // get number of zeroXs
37
38 ~zeroXNum = ~zeroXs.size
39
40
41 (
42 SynthDef(\gran_1g, { |out = 0, soundBuf, zeroXBuf, zeroXNum, posLo = 0.1, posHi =
0.9,
43   posRateE = 0, posRateM = 1, rate = 1, amp = 1|
44   var sig, bufDur, pos, posRate;
45
46   posRate = 10 ** posRateE * posRateM;
47   bufDur = BufDur.kr(soundBuf);
48
49   // move through the buffer of zero crossings
50   // the tempo (rate) refers to the sound buffer
51   pos = Phasor.ar(
52     0,
53     BufRateScale.kr(soundBuf) * posRate * SampleDur.ir / bufDur,
54     posLo,
55     posHi
56   ) * zeroXNum;
57
58   sig = ZeroXBufRd.ar(
59     soundBuf,
60     zeroXBuf,
61     0!2,
62     pos,
63     mul: amp,
64     rate: rate * [1, 1.01]
65   );
66   Out.ar(0, sig * amp)
67 }, metadata: {
68   specs: {
69     posLo: [0.01, 0.99, \lin, 0.01, 0.1],
70     posHi: [0.01, 0.99, \lin, 0.01, 0.5],
71     posRateE: [-2, 2, \lin, 1, -1],
72     posRateM: [0.1, 10, \exp, 0.01, 3],
73     rate: [0.3, 3, \lin, 0.01, 1],
74     amp: [0.0, 2, \lin, 0.005, 1]
75   }
76 }
77 ).add;
78
79 \gran_1g.sVarGui([\soundBuf, b.bufnum, \zeroXBuf, z.bufnum, \zeroXNum,
~zeroXNum]).gui
80 )

```

Ex.1h: Buffer granulation with (half) wavesets: TZeroXBufRd

```
1 // buffer preparations from Ex. 1g
```

```

2
3 // again a movement through the buffer of zero crossings is implemented
4 // the repetition of wavesets though is defined by xNum and xRep
5
6 // with high trigger rates, xNum and xRep values and low rates you might
7 // have to increase overlapSize, see TZeroXBufRd help for a discussion of this topic
8
9 (
10 SynthDef(\gran_1h, { |out = 0, soundBuf, zeroXBuf, zeroXNum, posLo = 0.1, posHi =
11 0.9,
12   posRateE = 0, posRateM = 1, xNum = 1, xRep = 1, rate = 1, trigRate = 100, amp =
13 1|
14   var sig, bufDur, pos, posRate;
15
16   posRate = 10 ** posRateE * posRateM;
17
18   bufDur = BufDur.kr(soundBuf);
19   pos = Phasor.ar(
20     0,
21     BufRateScale.kr(soundBuf) * posRate * SampleDur.ir / bufDur,
22     posLo,
23     posHi
24   ) * zeroXNum;
25
26   // move through the buffer of zero crossings
27   // the tempo (rate) refers to the sound buffer
28   sig = TZeroXBufRd.ar(
29     soundBuf,
30     zeroXBuf,
31     0!2, // bufMix arg triggers stereo
32     trig: Impulse.ar(trigRate),
33     zeroX: pos,
34     xNum: xNum,
35     xRep: xRep,
36     mul: amp,
37     rate: rate * [1, 1.01], // a bit of decorrelation
38     overlapSize: 20 // larger overlapSize
39   );
40   // by overlapping a DC can easily accumulate, so do leak
41   Out.ar(0, LeakDC.ar(sig) * amp)
42 }, metadata: (
43   specs: (
44     posLo: [0.01, 0.99, \lin, 0.01, 0.1],
45     posHi: [0.01, 0.99, \lin, 0.01, 0.5],
46     posRateE: [-2, 2, \lin, 1, -1],
47     posRateM: [0.1, 10, \exp, 0.01, 1],
48     rate: [0.3, 2, \lin, 0.01, 1],
49     trigRate: [5, 120, \lin, 0, 50],
50     xNum: [1, 5, \lin, 1, 2],
51     xRep: [1, 5, \lin, 1, 3],
52     amp: [0.0, 2, \lin, 0, 1]
53   )
54 )
55 ).add;
56 \gran_1h.sVarGui([\soundBuf, b.bufnum, \zeroXBuf, z.bufnum, \zeroXNum,
~zeroXNum]).gui
)

```

2. Granulation driven by language

Ex.2a: Basic buffer granulation Pbind

NOTE: Language-driven sequencing is not sample-exact in realtime (with NRT synthesis it is). This is related to hardware control and cannot be overcome currently. However for most practical purposes this might not be relevant. It is anyway a far less strong effect than the inaccuracies related to `Out.ar` and the combination of `OffsetOut.ar` and `In.ar` (see examples 2a-d in [Live Granulation](#))

```

1 // Control parameters like in Ex.1a, but implemented with a SynthDef for
2 // playing single grains and an appropriate Pbind,
3 // OffsetOut used for exact timing.
4
5 (
6 SynthDef(\gran_2a, { |out = 0, pos = 0, sndBuf = 0, windowBuf = 1, granDur = 0.1,
7   rate = 1, loop = 1, panMax = 0, amp = 1|
8   var window, src;
9   src = PlayBuf.ar(1, sndBuf, BufRateScale.kr(sndBuf) * rate,
10    1, round(pos * BufFrames.kr(sndBuf)), loop, 2);
11   window = BufRd.ar(1, windowBuf,
12     EnvGen.ar(Env([0, BufFrames.kr(windowBuf)], [granDur]),
13     doneAction: 2), loop, 4);
14   OffsetOut.ar(out, Pan2.ar(src, panMax, amp) * window);
15 }).add;
16
17 b = Buffer.read(s, Platform.miSCellaneousDirs[0] ++ "Sounds" ++
18 "kitchen_sounds_1.wav");
19 // This searches the most likely extension places for the miSCellaneous folder.
20 // In case of an extraordinary install situation or a removed sound file, pass the
21 // concerned path.
22
23 w = Buffer.sendCollection(s, Signal.hanningWindow(1024));
24
25 // Determining the correct buffer position depending on
26 // posRate, posLo and posHi needs a little calculation.
27 // In 1a this was done inside the ugen by a Phasor.
28 // See Ex.3b, 3c for doing position movement with a separate Synth.
29
30 // PL placeholder patterns used, could also be Pfunc { ~ ... }
31
32 (
33 p = Pbind(
34   \instrument, \gran_2a,
35   \sndBuf, b,
36   \windowBuf, w,
37
38   \dur, 1 / PL(\trigRate),
39   \granDur, PL(\granDur),
40   \time, Ptime(),
41   \pos, Pfunc { |e|
42     var relTime = ~posRate * e.time / e.sndBuf.duration, relDif;
43     relDif = ~posHi - ~posLo;
44     relTime + rand2(~posDev) % relDif + ~posLo;
45   },
46   \rate, PL(\rate),

```



```

47     \amp, PL(\amp),
48     \panMax, PLseq([-1,1]) * PL(\panMax),
49     \out, 0
50 );
51
52 VarGui([
53     \posLo, [0.0, 0.99, \lin, 0.01, 0],
54     \posHi, [0.0, 0.99, \lin, 0.01, 1],
55     \posRate, [0.1, 2, \lin, 0.01, 1],
56     \posDev, [0, 0.2, 5, 0, 0.01],
57     \trigRate, [1, 200, \lin, 0.01, 120],
58     \granDur, [0.01, 0.3, \lin, 0.005, 0.06],
59     \rate, [0.1, 3, \lin, 0.01, 1],
60     \panMax, [0.0, 1, \lin, 0.0, 0.8],
61     \amp, [0.0, 1, \lin, 0.01, 0.25]
62 ], stream: p
63 ).gui(varColorGroups: (0..8).clumps([4,1,1,1,1,1]))
64 )

```

Ex.2b: Switching between stochastic distributions

```

1 // Extended basic SynthDef from Ex.2a with bandpass filter
2
3 (
4 SynthDef(\gran_2b, { |out = 0, pos = 0, sndBuf = 0, windowBuf = 1, granDur = 0.1,
5   rate = 1, loop = 1, panMax = 0, amp = 1, bpFreq = 500, bpRQ = 0.5, bpWet = 1|
6   var window, granSrc, src;
7   granSrc = PlayBuf.ar(1, sndBuf, BufRateScale.kr(sndBuf) * rate,
8     1, round(pos * BuffFrames.kr(sndBuf)), loop, 2);
9   window = BufRd.ar(1, windowBuf,
10     EnvGen.ar(Env([0, BuffFrames.kr(windowBuf)], [granDur]),
11       doneAction: 2), loop, 4);
12   // do amplitude compensation, estimation like in Wavesets example by Alberto de
13   Campo
14   src = (BPF.ar(granSrc, bpFreq, bpRQ, mul: (bpRQ ** -1) * (400 / bpFreq ** 0.5))
15 *
16     bpWet + (granSrc * (1 - bpWet)));
17   OffsetOut.ar(out, Pan2.ar(src, panMax, amp) * window);
18 }).add;
19
20 b = Buffer.read(s, Platform.miSCellaneousDirs[0] ++ "Sounds" ++
21 "kitchen_sounds_1.wav");
22 // This searches the most likely extension places for the miSCellaneous folder.
23 // In case of an extraordinary install situation or a removed sound file, pass the
24 // concerned path.
25
26 w = Buffer.sendCollection(s, Signal.hanningWindow(1024));
27 )
28 // random types
29
30 // 0: low value
31 // 1: low or high value, evenly distributed
32 // 2: evenly distributed
33 // 3: linear decrease from mean value
34 // 4: exponential distribution

```

```

35 // 5: beta distribution, default parameter 0.3 centers value at the borders
36 // 6: brownian movement (of first order)
37 // 7: brownian movement of second order (stepsize itself generated by brownian
38 // movement)
39 // A second order brownian movement much more tends to get stuck at the
40 // borders than a normal (first order) brownian movement
41 // E.g. see Sergio Luque's presentation of Xenakis's stochastic synthesis:
42 // "Stochastic Synthesis, Origins and Extensions", pp 25-28
43 // http://sergioluque.com
44
45
46 // Function that generates an array of PLx patterns
47 // of different random types (see PLx suite).
48 // These placeholders can refer to environmental variables
49 // to be set by the VarGui interface later on.
50
51 (
52 d = { |keyLo, keyHi, betaProb = 0.3, brownStepFac = 0.01,
53       brown2Ratio = 1, brown2StepFac = 0.01|
54       // keyLo and keyHi must be Symbols,
55       // other args may be Symbols
56       var patLo, patHi, patDif;
57
58       // avoid lo-hi reversing with Pbrown
59       patLo = min(PL(keyLo), PL(keyHi));
60       patHi = max(PL(keyLo), PL(keyHi));
61       patDif = patHi - patLo;
62       [
63         PL(keyLo),
64         Pfunc { currentEnvironment[[keyLo, keyHi].choose] },
65         PLwhite(keyLo, keyHi),
66         PLmeanrand(keyLo, keyHi),
67         PLexprand(keyLo, keyHi),
68         PLbeta(keyLo, keyHi, betaProb, betaProb),
69         PLbrown(patLo, patHi, patDif * PL(brownStepFac)),
70         PLbrown(patLo, patHi,
71                 PLbrown(
72                     patDif.neg * PL(brown2Ratio) / 2,
73                     patDif * PL(brown2Ratio) / 2,
74                     patDif * PL(brown2Ratio) * PL(brown2StepFac)
75                 )
76             )
77         ]
78     };
79 )
80
81 // trigrate, granDur, rate and bpFreq are
82 // chosen between bounds according to the
83 // random distribution type notated with suffix D
84
85 // single grains are filtered with a bandpass
86 // amount of effect controlled with bpWet
87
88 (
89 p = Pbind(
90     \instrument, \gran_2b,
91     \sndBuf, b,
92     \windowBuf, w,
93
94     \dur, 1 / PLswitch1(d.( \trigRateLo, \trigRateHi), \trigRateD),
95     \granDur, PLswitch1(d.( \granDurLo, \granDurHi), \granDurD),
96     \time, Ptime(),
97     \posRate, PL(\posRate),

```

```

98   \pos, Pfunc { |e|
99     var relTime = ~posRate * e.time / e.sndBuf.duration, relDif;
100    relDif = ~posHi - ~posLo;
101    relTime + rand2(~posDev) % relDif + ~posLo;
102  },
103  \rate, PLswitch1(d.\rateLo, \rateHi), \rateD),
104  \bpFreq, PLswitch1(d.\bpFreqLo, \bpFreqHi), \bpFreqD),
105  \bpRQ, PL(\bpRQ),
106  \bpWet, PL(\bpWet),
107
108  \amp, PL(\amp),
109  \panMax, PLseq([-1,1]) * PL(\panMax),
110  \out, 0
111 );
112
113 VarGui([
114   \posLo, [0.0, 0.99, \lin, 0.01, 0.21],
115   \posHi, [0.0, 0.99, \lin, 0.01, 0.47],
116   \posRate, [0.1, 2, \lin, 0.01, 0.2],
117   \posDev, [0, 0.2, 5, 0, 0.002],
118
119   \trigRateLo, [1, 200, \lin, 0.01, 21],
120   \trigRateHi, [1, 200, \lin, 0.01, 155],
121   \trigRateD, [0, 7, \lin, 1, 6],
122
123   \granDurLo, [0.01, 0.6, \exp, 0.0, 0.037],
124   \granDurHi, [0.01, 0.6, \exp, 0.0, 0.4],
125   \granDurD, [0, 7, \lin, 1, 6],
126
127   \rateLo, [0.1, 3, \lin, 0.01, 1.09],
128   \rateHi, [0.1, 3, \lin, 0.01, 1.63],
129   \rateD, [0, 7, \lin, 1, 1],
130
131   \bpFreqLo, [50, 10000, \exp, 0.1, 54],
132   \bpFreqHi, [50, 10000, \exp, 0.1, 8275],
133   \bpFreqD, [0, 7, \lin, 1, 1],
134   \bpRQ, [0.01, 0.99, \lin, 0.0, 0.07],
135   \bpWet, [0.0, 1, \linear, 0.0, 0.23],
136
137   \panMax, [0.0, 1, \lin, 0.0, 0.85],
138   \amp, [0.0, 1, \lin, 0.01, 0.25]
139 ], stream: p
140 ).gui(varColorGroups: (0..19).cLumps([4,3,3,3,5,1,1]))
141 )

```

Ex.2c: Generating granular phrases with Pspawner

```

1 // This example needs SynthDef \gran_2b and Function d to be taken from Ex. 2b
2
3 (
4 b = Buffer.read(s, Platform.miSCellaneousDirs[0] ++ "Sounds" ++
  "kitchen_sounds_1.wav");
5 // This searches the most likely extension places for the miSCellaneous folder.
6 // In case of an extraordinary install situation or a removed sound file, pass the
  concerned path.
7
8 w = Buffer.sendCollection(s, Signal.hanningWindow(1024));
9 )

```

```

10
11 // A simple form of Pspawner is used to generate phrases.
12 // Phrase length params are taken a bit roughly as sustain and
13 // rest times also depend on randomly varying grain lengths.
14 // spSustain controls medium sustain time (without grain length overhead)
15 // spLegato controls medium legato factor (disregarding reduction by grain length
16 // overhead)
17 // spDev is causing separate random deviation of spSustain and spLegato
18 // between 1/(1+spDev) and 1+spDev
19
20 // random distribution switching is restricted here to
21 // types 6 and 7 (random walks of first and second order)
22 // to force individual sound qualities of phrases.
23
24 (
25 // declare var here as pattern is repeatedly evaluated from within the Pspawner,
26 // interpreter variable would be unsafe if running examples in parallel
27
28 var p = Pbind(
29   \instrument, \gran_2b,
30   \sndBuf, b,
31   \windowBuf, w,
32
33   \dur, 1 / PLswitch1(d.\trigRateLo, \trigRateHi), \trigRateD),
34   \granDur, PLswitch1(d.\granDurLo, \granDurHi), \granDurD),
35   \time, Ptime(),
36   \posRate, PL(\posRate),
37
38   // random timeOffset added with each spawning
39   \pos, Pfunc { |e|
40     var relTime = ~posRate * e.time / e.sndBuf.duration + e.timeOffset, relDif;
41     relDif = ~posHi - ~posLo;
42     relTime + rand2(~posDev) % relDif + ~posLo;
43   },
44   \rate, PLswitch1(d.\rateLo, \rateHi), \rateD),
45   \bpFreq, PLswitch1(d.\bpFreqLo, \bpFreqHi), \bpFreqD),
46   \bpRQ, PL(\bpRQ),
47   \bpWet, PL(\bpWet),
48
49   \amp, PL(\amp),
50   \panMax, PLseq([-1,1]) * PL(\panMax),
51   \out, 0
52 );
53
54 q = Pspawner({ |sp|
55   var randomizer = { |x| var y = rand(x); 0.5.coin.if { 1 + y } { 1 / (1 + y) } },
56   sus, legato, delta;
57
58   loop {
59     sus = ~spSustain * randomizer.(~spDev);
60     legato = ~spLegato * randomizer.(~spDev);
61
62     // take random offset for each phrase
63     sp.par(Pfindur(sus, Psetpre(\timeOffset, 5.0.rand, p)));
64     delta = sus / (~spLegato * randomizer.(~spDev));
65     sp.wait(delta)
66   }
67 });
68
69 VarGui([
70   \posLo, [0.0, 0.99, \lin, 0.01, 0.16],
71   \posHi, [0.0, 0.99, \lin, 0.01, 0.41],
72   \posRate, [0.1, 2, \lin, 0.01, 1.4],
73   \posDev, [0, 0.2, 5, 0, 0.0017],

```

```

73
74 \trigRateLo, [1, 200, \lin, 0.01, 70],
75 \trigRateHi, [1, 200, \lin, 0.01, 150],
76 \trigRateD, [6, 7, \lin, 1, 7],
77
78 \granDurLo, [0.01, 0.6, \exp, 0.0, 0.02],
79 \granDurHi, [0.01, 0.6, \exp, 0.0, 0.11],
80 \granDurD, [6, 7, \lin, 1, 6],
81
82 \rateLo, [0.1, 3, \lin, 0.01, 0.2],
83 \rateHi, [0.1, 3, \lin, 0.01, 1.86],
84 \rateD, [6, 7, \lin, 1, 7],
85
86 \bpFreqLo, [50, 10000, \exp, 0.1, 67],
87 \bpFreqHi, [50, 10000, \exp, 0.1, 5885],
88 \bpFreqD, [6, 7, \lin, 1, 6],
89 \bpRQ, [0.01, 0.99, \lin, 0.0, 0.17],
90 \bpWet, [0.0, 1, \linear, 0.0, 0.38],
91
92 \spSustain, [0.2, 2, \linear, 0.0, 0.884],
93 \spLegato, [0.6, 1.2, \linear, 0.0, 0.996],
94 \spDev, [0.0, 1, \linear, 0.0, 0.41],
95
96 \panMax, [0.0, 1, \lin, 0.0, 0.85],
97 \amp, [0.0, 1, \lin, 0.01, 0.35]
98 ], stream: q
99 ).gui(varColorGroups: (0..22).clumps([4,3,3,3,5,3,1,1]) )
100 )

```

Ex.2d: Wave sets

```

1 // For this example you need Alberto de Campo's Wavesets class
2 // (Quark extension, implementation following definitions of Trevor Wishart)
3 // and the Function d from Ex.2b.
4
5 // the wave set player synth optionally adds a BPF applied to the signal,
6 // amount can be controlled with bpWet
7 // attack and release time > 0 for smoothening
8
9 (
10 SynthDef(\wsPlayer, { arg out = 0, buf = 0, start = 0, length = 441,
11   rate = 1, att = 0.03, rel = 0.03, wvDur = 1, panMax = 0, amp = 0.2,
12   delayL = 0.0, delayR = 0.0, bpFreq = 500, bpRQ = 0.5, bpWet = 1;
13   var phasor, env, granSrc, src, attEff, relEff, sus;
14
15   phasor = Phasor.ar(0, BufRateScale.ir(buf) * rate, 0, length) + start;
16   attEff = min(att, wvDur/2);
17   relEff = min(rel, wvDur/2);
18   sus = wvDur - attEff - relEff;
19
20   env = EnvGen.ar(Env([0, 1, 1, 0], [attEff, sus, relEff], \sine), doneAction: 2);
21   granSrc = BufRd.ar(1, buf, phasor);
22   src = (BPF.ar(granSrc, bpFreq, bpRQ, mul: (bpRQ ** -1) * (400 / bpFreq ** 0.5))
23 *
24   bpWet + (granSrc * (1 - bpWet)));
25   OffsetOut.ar(out, Pan2.ar(src, panMax, amp) * env);
26 }).add;

```

```

26
27 a = Buffer.read(s, Platform.miSCellaneousDirs[0] ++ "Sounds" ++
    "kitchen_sounds_1.wav");
28 // This searches the most likely extension places for the miSCellaneous folder.
29 // In case of an extraordinary install situation or a removed sound file, pass the
    concerned path.
30
31 b = Buffer.read(s, a);
32 w = Wavesets.from(a);
33
34 // relative positions of zero crossings
35 x = w.fracXings.drop(-1) / w.numFrames;
36 )
37
38
39 // Note that trigRate / event duration is not controlled directly in this setup,
40 // it's derived from wave set length (may vary) and legato factor.
41 // For short wave sets and a low legato value durs could become so short that
42 // a sudden mass of grains could cause hangs.
43 // To avoid this a parameter maxTrigRate is invented.
44
45 // As wavesets start at distinguished positions in the buffer and
46 // - together with legato - durations are determined,
47 // an additionally given position rate in general cannot result in a "correct"
48 // looping through the buffer, there will be a deviation.
49 // Here two types of approximation can be choosen with posType:
50 // Type 0 takes the waveset nearest to the calculated exact position.
51 // Type 1 linearly maps positions to wave set indices.
52 // As wave sets are of different length the latter is a rough heuristic
53 // accelerating buffer parts with relatively low frequencies.
54
55 (
56 p = Pbind(
57     \instrument, \wsPlayer,
58
59     // referring to interpreter variables within Pfuncs
60     // when running several examples in parallel is unsafe,
61     // so pass them here, then access from within the event
62
63     \b, b,
64     \w, w,
65     \x, x,
66
67     \time, Ptime(),
68     \posLo, PL(\posLo),
69     \posHi, PL(\posHi),
70     \posRate, PL(\posRate),
71
72     \pos, Pfunc { |e| (e.time * e.posRate / e.b.duration) %
73         (e.posHi - e.posLo) + e.posLo },
74
75     // Estimation of ws index from relative position, see explanation above,
76     // indexIn might be a bottleneck with large buffers,
77     // also a more rough estimation of ws index could be used:
78     // \startWs, Pfunc { |e| e.pos.linlin(0, 1, 0, e.w.xings.size -
79     2).round.asInteger }
80
81     \startWs, Pfunc { |e|
82         (~posType == 0).if {
83             e.x.indexIn(e.pos)
84         }{
85             e.pos.linlin(e.posLo, e.posHi, e.x.indexIn(e.posLo),
86                 e.x.indexIn(e.posHi))
87             .round.asInteger
88         }
89     }
90 )

```

```

86     }
87     },
88
89     \numWs, PLswitch1(d.\numWsLo, \numWsHi), \numWsD),
90     \repeats, PLswitch1(d.\repeatsLo, \repeatsHi), \repeatsD),
91
92     \bpFreq, PLswitch1(d.\bpFreqLo, \bpFreqHi), \bpFreqD),
93     \bpRQ, PL(\bpRQ),
94     \bpWet, PL(\bpWet),
95
96     \rate, PLswitch1(d.\rateLo, \rateHi), \rateD),
97     \data, Pfunc { |e| e.w.frameFor(e.startWs, e.numWs) },
98
99     \buf, b.bufnum,
100    \start, Pkey(\data).collect(_[0]), // startFrame
101    \length, Pkey(\data).collect(_[1]), // length (frameNum)
102    \wvDur, Pkey(\data).collect(_[2]) * Pkey(\repeats), // sustain time
103
104    \calculatedDur, Pkey(\wvDur) / PLswitch1(d.\legatoLo, \legatoHi), \legatoD),
105    \dur, Pfunc { |e| max(e.calculatedDur, 1 / ~maxTrigRate) },
106
107    \panMax, PLseq([-1,1]) * PL(\panMax),
108    \amp, PL(\amp),
109    \out, 0
110 );
111
112
113 VarGui([
114     \posLo, [0, 1, \lin, 0.0, 0.15],
115     \posHi, [0, 1, \lin, 0.0, 0.45],
116     \posRate, [0.0, 2, \lin, 0.01, 0.25],
117     \posType, [0, 1, \lin, 1, 0],
118
119     \numWsLo, [1, 100, \lin, 1, 5],
120     \numWsHi, [1, 100, \lin, 1, 23],
121     \numWsD, [0, 7, \lin, 1, 6],
122
123     \repeatsLo, [1, 4, \lin, 1, 1],
124     \repeatsHi, [1, 4, \lin, 1, 2],
125     \repeatsD, [0, 7, \lin, 1, 6],
126
127     \maxTrigRate, [1, 250, \lin, 1, 200],
128
129     \bpFreqLo, [50, 10000, \exp, 0.1, 67],
130     \bpFreqHi, [50, 10000, \exp, 0.1, 9600],
131     \bpFreqD, [0, 7, \lin, 1, 7],
132     \bpRQ, [0.01, 0.99, \lin, 0.005, 0.22],
133     \bpWet, [0.0, 1, \lin, 0.005, 0.0],
134
135     \rateLo, [0.05, 2, \lin, 0.0, 0.32],
136     \rateHi, [0.05, 2, \lin, 0.0, 1.3],
137     \rateD, [0, 7, \lin, 1, 7],
138
139     \att, [0.0, 0.05, \lin, 0.001, 0.001],
140     \rel, [0.0, 0.05, \lin, 0.001, 0.001],
141
142     \panMax, [0.0, 1, \lin, 0, 0.8],
143     \legatoLo, [0.3, 25, \exp, 0, 0.4],
144     \legatoHi, [0.3, 25, \exp, 0, 5.5],
145     \legatoD, [0, 7, \lin, 1, 7],
146     \amp, [0.0, 2.0, \lin, 0.0, 0.7]
147 ],
148     stream: p
149 ).gui(varColorGroups: (0..25).clumps([4,3,3,1,5,3,2,1,3,1]))

```


150)

3. Hybrid Implementations

Ex.3a: Granulation with ugen plus step sequencing

```

1 // Here the trigger for the TGrains ugen comes from a Pbind
2 // which also generates rates like a step sequencer
3
4
5 (
6 SynthDef(\gran_3a, { arg out = 0, posLo = 0.0, posHi = 1.0,
7   posRate = 1, posDev = 0.01, bufNum = 0, t_trig = 0,
8   granDur = 0.1, t_rate = 1.0, rateDev = 0,
9   panMax = 1, amp = 0.1, interp = 4;
10
11   var pan, pos, bufDur, bufDurSection, posDif;
12
13   posDif = posHi - posLo;
14   bufDur = BufDur.kr(bufNum);
15   bufDurSection = bufDur * posDif;
16   pos = posLo * bufDur +
17     (Phasor.ar(0, BufRateScale.kr(bufNum) * posRate / SampleRate.ir, posLo *
18   bufDur, posHi * bufDur) +
19     (TRand.kr(-0.5, 0.5, t_trig) * posDev * bufDur)).mod(bufDurSection);
20   pan = Demand.kr(t_trig, 0, Dseq([panMax, panMax.neg], inf) * 0.999);
21   Out.ar(out, TGrains.ar(2, t_trig, bufNum, t_rate, pos, granDur, pan, 1, interp) *
22   amp);
23   }, metadata: (
24     specs: (
25       posLo: [0.01, 0.99, \lin, 0.01, 0],
26       posHi: [0.01, 0.99, \lin, 0.01, 1],
27       posRate: [0.1, 2, \lin, 0.01, 1],
28       posDev: [0, 0.2, 5, 0, 0.01],
29       panMax: [0.0, 1, \lin, 0.005, 0.8],
30       amp: [0.0, 1, \lin, 0.005, 0.5]
31     )
32   )
33 ).add;
34
35 b = Buffer.read(s, Platform.miSCellaneousDirs[0] ++ "Sounds" ++
36 "kitchen_sounds_1.wav");
37
38
39 // This searches the most likely extension places for the miSCellaneous folder.
40 // In case of an extraordinary install situation or a removed sound file, pass the
41 // concerned path.
42 )
43
44 // As the setting Pbind needs to know the Synth's nodeID
45 // the Synth has to be started explicitly and passed to the VarGui later on
46 // (VarGui takes Synths as well as SynthDefs, passing a SynthDef is recommended in
47 // general).
48 // The Synth starts silently as t_trig defaults to 0.
49
50 (

```

```

46 x = Synth(\gran_3a, [\bufNum, b]).register;
47
48 p = Pbind(
49   \type, \set,
50   \id, x,
51   \args, [\t_trig, \t_rate, \granDur],
52
53   \dur, PL(\dur),
54   \granDur, Pkey(\dur) * PL(\legato),
55   \t_trig, 1,
56   \t_rate, PLseq(\midi).midiratio
57 );
58 )
59
60 // Do start and pause with the Pbind (EventStreamPlayer) player.
61 // If you stop the Synth you cannot resume audio with a new Synth
62 // as the EventStreamPlayer has lost the correct nodeID
63 // (however the Synth can be paused and resumed).
64
65 (
66   VarGui(varCtr: [
67     \dur, [0.01, 0.1, \lin, 0, 0.05],
68     \legato, [0.3, 3, \lin, 0, 1],
69     \midi, [-12, 12, \lin, 1, 1] ! 8
70   ], synth: x, stream: p
71 ).gui;
72 )

```

Ex.3b: Using external control synths

```

1 // Example needs SynthDef from Ex.2a
2
3 // Also with language-driven granulation
4 // controls can be delegated to separate Synths,
5 // which output to control buses.
6 // Control inputs of single synths can read from
7 // these buses (comfortably use aBus.asMap in the Pbind)
8 // or synths can read from buses with In.kr (needs extra definition).
9
10 // A nearby parameter to determine with a separate synth is grain position
11
12 (
13   SynthDef(\bufPhasor, { |out = 0, sndBuf = 0, posRate = 1, posLo = 0, posHi = 1,
14     posDev = 0.01|
15     var pos, posDif;
16     posDif = posHi - posLo;
17     pos = Phasor.ar(1, posRate * BufRateScale.kr(sndBuf) / BufFrames.kr(sndBuf), 0,
18     posDif)
19     + WhiteNoise.kr(posDev / 2) % posDif + posLo;
20     Out.kr(out, A2K.kr(pos));
21   }
22 ).add;
23
24 b = Buffer.read(s, Platform.miSCellaneousDirs[0] ++ "Sounds" ++
25 "kitchen_sounds_1.wav");
26 // This searches the most likely extension places for the miSCellaneous folder.
27 // In case of an extraordinary install situation or a removed sound file, pass the
28 // concerned path.
29
30 w = Buffer.sendCollection(s, Signal.hanningWindow(1024));
31 c = Bus.control(s,1);
32 )

```

```

29
30 // in GUI start EventStreamPlayer and bufPhasor Synth
31
32 (
33 p = Pbind(
34   \instrument, \gran_2a,
35   \sndBuf, b,
36   \windowBuf, w,
37
38   \dur, 1 / PL(\trigRate),
39   \granDur, PL(\granDur),
40
41   \pos, c.asMap,
42   \rate, PL(\rate),
43   \amp, PL(\amp),
44   \panMax, PLseq([-1,1]) * PL(\panMax),
45   \out, 0
46 );
47
48 VarGui([
49   \trigRate, [1, 200, \lin, 0.01, 50],
50   \granDur, [0.01, 0.3, \lin, 0.005, 0.12],
51   \rate, [0.1, 3, \lin, 0.01, 1],
52   \panMax, [0.0, 1, \lin, 0.0, 0.8],
53   \amp, [0.0, 1, \lin, 0.01, 0.3]
54 ], [
55   \out, c.index,
56   \sndBuf, b.bufnum,
57   \posLo, [0, 1, \linear, 0.005, 0],
58   \posHi, [0, 1, \linear, 0.005, 1],
59   \posRate, [0.1, 2, \linear, 0.01, 1],
60   \posDev, [0, 0.2, 5, 0, 0.01]
61 ],
62 p, \bufPhasor
63 ).gui(sliderPriority: \synth, playerPriority: \synth);
64 )

```

Ex.3c: Switching between ugens of external control synths

```

1 // Example needs SynthDef from Ex.2a
2
3 // external control synth for position as in Ex.3a, but with
4 // different types of movement to select
5
6 (
7 SynthDef(\bufPosLFO, { |out = 0, lfoType = 0, freq = 1, posLo = 0, posHi = 1, posDev
8 = 0.01|
9   var pos, posDif;
10  posDif = posHi - posLo;
11  pos = WhiteNoise.kr(posDev / 2) + Select.kr(lfoType,
12    [LFDNoise0, LFDNoise1, LFDNoise3].collect(_.kr(freq, posDif))) % posDif +
13  posLo;
14  Out.kr(out, pos);
15 }).add;
16
17 b = Buffer.read(s, Platform.miSCellaneousDirs[0] ++ "Sounds" ++
18 "kitchen_sounds_1.wav");
19 // This searches the most likely extension places for the miSCellaneous folder.
20 // In case of an extraordinary install situation or a removed sound file, pass the
21 // concerned path.
22
23 w = Buffer.sendCollection(s, Signal.hanningWindow(1024));

```

```

20 c = Bus.control(s,1);
21 )
22
23
24 // added parallel grains
25 // in GUI start bufPhasor Synth before or together with EventStreamPlayer
26
27 // buffer position movement can be forward and backward
28 // lfoType 0: LFDNoise0 (jumps)
29 // lfoType 1: LFDNoise1 (linear interpolation)
30 // lfoType 2: LFDNoise3 (cubic interpolation, smooth, useful in many control
    contexts)
31
32 (
33 p = Pbind(
34   \instrument, \gran_2a,
35   \sndBuf, b,
36   \windowBuf, w,
37
38   \dur, 1 / PL(\trigRate),
39   \granDur, PL(\granDur),
40
41   \pos, c.asMap,
42   \rate, PL(\rate) * PL(\midiAdd).midiratio,
43   \amp, PL(\amp),
44   \panMax, PLseq([-1,1]) * PL(\panMax),
45   \out, 0
46 );
47
48 VarGui([
49   \trigRate, [1, 200, \lin, 0.01, 80],
50   \granDur, [0.01, 0.3, \lin, 0.005, 0.195],
51   \rate, [0.1, 1.5, \lin, 0.01, 0.6],
52   \midiAdd, [-5, 0].collect([-12, 12, \lin, 1, _]),
53   \panMax, [0.0, 1, \lin, 0.0, 0.95],
54   \amp, [0.0, 1, \lin, 0.01, 0.15]
55 ],[
56   \out, c.index,
57   \posLo, [0, 1, \linear, 0.005, 0.15],
58   \posHi, [0, 1, \linear, 0.005, 0.43],
59   \posDev, [0, 0.2, 5, 0, 0.0],
60   \freq, [0.01, 2, \lin, 0, 0.96],
61   \lfoType, [0, 2, \lin, 1, 2]
62 ],
63 p, \bufPosLFO
64 ).gui(sliderPriority: \synth, playerPriority: \synth);
65 )

```

Ex.3d: Pattern-driven sequencing of granular events using demand rate ugens

```

1 // This is basically the same as SynthDef \gran_1d with an additional envelope,
2 // grain position phasor is left out, position is determined by a phasor synth via
    bus and mapping.
3
4 (
5 // length of demand rate sequence, you might want to check larger sizes
6 // in connection with gui arg tryColumnNum > 1
7 ~n = 5;
8
9

```

```

10 SynthDef(\gran_3d, { |out = 0, soundBuf, envBuf, att = 0.1, sus = 1, rel = 1, pos =
11 0.5,
12   granDurMul = 1, rateMul = 1, overlapMul = 1, panMul = 1, amp = 0.5, interp = 2|
13   var signal, bufDur, granDur, granGate, relGranDurs, overlap, relOverlaps,
14   overlapSeq,
15   pan, rate, relRates, rateSeq, granDurSeq;
16   // array args for demand rate sequencing, short form of NamedControl
17   relGranDurs = \relGranDurs.kr(0.1!~n);
18   relRates = \relRates.kr(1!~n);
19   relOverlaps = \relOverlaps.kr(1!~n);
20
21   // Dstutter (or Dunique) necessary as granDurSeq is polled twice: granGate and
22   granDur
23   granDurSeq = Dstutter(2, Dseq(relGranDurs, inf));
24   rateSeq = Dseq(relRates, inf);
25   overlapSeq = Dseq(relOverlaps, inf);
26
27   granGate = TDuty.ar(granDurSeq * granDurMul);
28   granDur = Demand.ar(granGate, 0, granDurSeq * granDurMul);
29   rate = Demand.ar(granGate, 0, rateSeq) * rateMul;
30   pan = Demand.ar(granGate, 0, Dseq([1, -1], inf)) * 0.999 * panMul;
31   overlap = Demand.ar(granGate, 0, overlapSeq) * overlapMul;
32
33   bufDur = BufDur.kr(soundBuf);
34   signal = TGrains.ar(2, granGate, soundBuf, rate, pos * bufDur, granDur *
35   overlap, pan, 1, interp) *
36   EnvGen.kr(Env([0, 1, 1, 0], [att, sus, rel]), doneAction: 2);
37
38   Out.ar(out, signal * amp);
39 }
40 ).add;
41
42 SynthDef(\bufPhasor_2, { |out = 0, sndBuf = 0, posRateE = 0, posRateM = 1,
43   posLo = 0, posHi = 1, posDev = 0.01|
44
45   var pos, posDif, posRate;
46   posDif = posHi - posLo;
47   posRate = 10 ** posRateE * posRateM;
48   pos = Phasor.ar(1, posRate * BufRateScale.kr(sndBuf) / BufFrames.kr(sndBuf), 0,
49   posDif)
50   + WhiteNoise.kr(posDev / 2) % posDif + posLo;
51   Out.kr(out, A2K.kr(pos));
52 }
53 ).add;
54
55 b = Buffer.read(s, Platform.miSCellaneousDirs[0] +/+ "Sounds" +/+
56 "kitchen_sounds_1.wav");
57 // This searches the most likely extension places for the miSCellaneous folder.
58 // In case of an extraordinary install situation or a removed sound file, pass the
59 // concerned path.
60
61 c = Bus.control(s, 1);
62 )
63
64 // Pattern control of granular events
65
66 // Every granular event gets a duration between durLo and durHi (exp distribution)
67 // and an envelope according to att, sus and rel.
68 // One of four events is randomly defined as rest.
69
70 // arg arrays relGranDurs, relRates and relOverlaps determine

```

```

67 // demand rate sequencing for granulation as in Ex. 1d.
68 // Per event they are multiplied with corresponding factors
69 // limited by granDurMulLo/Hi, rateMulLo/Hi and overlapMulLo/Hi
70
71 // start phasor synth (first row in player console) before event stream player
72 // event stream player might start with rest
73
74
75 (
76 // Passing arrayed args with an event pattern requires
77 // wrapping them into an array or Ref object.
78 // This is necessary to distinguish from triggering
79 // several synths per event.
80 // .collect(`_) is short for .collect { |x| Ref(x) }
81 // .collect([_]) or .collect { |x| [x] } would also be possible
82
83 p = Pbind(
84   \instrument, \gran_3d,
85   \soundBuf, b,
86   \pos, c.asMap,
87
88   \dur, Plexprand(\durLo, \durHi),
89   \type, PLshufn(\note!3 ++ \rest),
90   \att, PL(\att),
91   \sus, PL(\sus),
92   \rel, PL(\rel),
93
94
95   \relGranDurs, PL(\relGranDurs).collect(`_),
96   \granDurMul, PLwhite(\granDurMulLo, \granDurMulHi),
97
98   \relOverlaps, PL(\relOverlaps).collect(`_),
99   \overlapMul, PLwhite(\overlapMulLo, \overlapMulHi),
100
101   \relRates, PL(\relRates).collect(`_),
102   \rateMul, PLwhite(\rateMulLo, \rateMulHi),
103
104   \panMul, PL(\panMul),
105   \amp, PL(\amp)
106 );
107
108 VarGui([
109   \durLo, [0.05, 1.5, \lin, 0, 0.16],
110   \durHi, [0.05, 1.5, \lin, 0, 1.2],
111   \att, [0.01, 0.6, \lin, 0, 0.4],
112   \sus, [0.01, 0.6, \lin, 0, 0.05],
113   \rel, [0.01, 0.6, \lin, 0, 0.5],
114   \relGranDurs, { |i| [0.01, 0.1, \lin, 0, i * 0.005 + 0.02] } ! ~n,
115   \granDurMulLo, [0.03, 2, \lin, 0, 0.05],
116   \granDurMulHi, [0.03, 2, \lin, 0, 1.8],
117
118   \relRates, { |i| [0.1, 1.5, \lin, 0, (5-i) * 0.1 + 0.5] } ! ~n,
119   \rateMulLo, [0.1, 2, \lin, 0, 0.5],
120   \rateMulHi, [0.1, 2, \lin, 0, 1.5],
121
122   \relOverlaps, [0.5, 3, \lin, 0, 2] ! ~n,
123   \overlapMulLo, [0.1, 2, \lin, 0, 0.25],
124   \overlapMulHi, [0.1, 2, \lin, 0, 1.8],
125
126   \panMul, [0.0, 1, \lin, 0.0, 0.8],
127   \amp, [0.0, 3, \lin, 0.01, 1.8]
128 ], [
129   \out, c.index,
130   \sndBuf, b.bufnum,

```

```

131     \posLo, [0, 1, \lin, 0.005, 0.1],
132     \posHi, [0, 1, \lin, 0.005, 0.9],
133     \posRateE, [-3, 4, \lin, -1, 0],
134     \posRateM, [0.1, 10, \exp, 0.01, 1],
135     \posDev, [0, 0.2, 5, 0, 0.01]
136   ],
137   p, \bufPhasor_2
138 ).gui(
139   tryColumnNum: 2,
140   sliderPriority: \synth,
141   playerPriority: \synth,
142   varColorGroups: (0..(~n*3+12)).clumps([2,3,~n+2,~n+2,~n+2,1,1]),
143   synthColorGroups: (0..6).clumps([1,1,2,2,1]),
144   labelWidth: 90,
145   sliderWidth: 300
146 );
147 )

```

4. Extensions of Setups

- **Changing ranges and scaling**

This concerns all control parameters in question. E.g. the layering of long grains (> 200 ms) in connection with small rates of position changes (posRate) often has interesting effects. One may want to drop the term granulation in that case, though it's the same structure of synthesis. For such parameters with a very large coefficient boundHi / boundLo one may take exponential scaling, and if this is not fine enough you can invent a control pair of mantissa and exponent (as for posRate in [Ex.1b](#)).

- **Parameter linkage**

On the one hand a logical restriction, on the other hand it can make sense from a musical / perceptual point of view. E.g. shortening of grains could be linked with a raise of rate (as low frequencies might fail to unfold in short grains). Anyway parameter linkage is reducing complexity - it's a trade-off between simplicity of the interface and exclusion of certain constellations which should be considered from case to case.

- **Inventing and extending controls and LFO changes dependant on specific buffers and parameter sets**

Say one has started playing around with a certain buffer and a general granulation patch. A parameter set that gives an interesting sound may react in a very interesting way on a change of a single parameter e.g., trigger rate. Then it may be an option to build in a control or a LFO specifically designed for that parameter - LFO is meant here in a general sense, it could be a LFO in a Synth, a dedicated LFO synth or defined by a rapidly sequencing Pattern.

- **Iterated granulation**

Granulation of buffers themselves resulting from buffer granulation can give interesting effects.

- **Spatialization**

For the sake of ease and comparison a L/R-switch per grain with one panning parameter was used in the examples above. Needless to say that spatial scattering of grains remains a large field of experimentation. Generally spoken spatialization can be part of the synthesis process or carried through independently afterwards, but also a combination of both approaches is feasible.

- **Effects**

Effect processing can be applied to all or single grains in a language-driven granulation setup. There can be one or more effects, serial or in parallel, defined outside or inside the Synth playing the buffer, as with the BPF in [Ex.2b](#), [Ex.2c](#), [Ex.2d](#). In these examples the bandpass is applied in general, but also a sequencing of effects can be done. And even in the case of just one effect (e.g. reverb) a decent sequencing of Fx- / noFx-events can sound very interesting. This can be done with PbindFx (as in the project [kitchen studies](#)), sequencing not more than one effect per grain can be done straightly: continously running effect synths read from different buses and events with different out values cause the player synths to output to these buses.

- **Micro rhythm**

See Xenakis' suggestion of Sieves with rhythm generation as a special application, a granulation example is contained in the example section of: [Sieves and Psieve patterns](#). PSPdiv, a dynamic multi-layer pulse divider, which is based on Pspawner, can also be used in this context, see the last example of [PSPdiv](#).